

Functions

You cannot do very much in ActionScript without using functions. Anytime you have an eventListener, you must write a function for the event to call. All of the functions that are called when an event occurs receive the type of event as an argument. In this lesson we will look at some of the other uses for functions.

MovieClips as Arguments

In this program we will have a face and a movie clip called “tie” with a different tie in each frame. When the user clicks on the tie, the tie moves to the next frame.

- Start a new Flash ActionScript3 file and save it as tie fla.
- Draw a face on the stage.
- On another layer draw a bow tie on the face.
- Select the bow tie and make it a movie clip called Tie.
- Add several frames to the tie clip, each a different color or size tie.
- On the stage, name the movie clip tie.
- Write the code as shown below and test the movie.

```
tie.stop(); //so that the tie will not play each frame
tie.buttonMode=true; //show a hand when you move mouse over it;
tie.addEventListener(MouseEvent.CLICK,nextTie);
function nextTie(e:MouseEvent):void { //show the next tie
    var f:int=tie.currentFrame+1;    //f is next frame
    if(f>tie.totalFrames) f=1;        //already at end, go back to frame 1
    tie.gotoAndStop(f);               //show the next tie
} //nextTie
```

Now suppose you want to add a hat that you can click to see all of the different hats. You could copy and paste all of the code above, then change the word tie to hat. Instead we will use one function to change either the hat or the tie. We can do this by using `e.target` instead of tie in the function. We will also write a function `initMC` that receives a movie clip as an argument. Then we can call that function and pass it tie, then call it again and pass hat:

```
initMC(tie);
initMC(hat);
function initMC(mc:MovieClip):void {
    mc.stop(); //so that the clip will not play each frame
    mc.buttonMode=true; //show a hand when you move mouse over it;
    mc.addEventListener(MouseEvent.CLICK,nextPic);
} //initMC
function nextPic(e:MouseEvent):void {
    var f:int=e.target.currentFrame+1; //f is next frame
    if(f>e.target.totalFrames) f=1; //already at end, go back to frame 1
    e.target.gotoAndStop(f); //show the next frame of e.target
} //nextPic
```

Experiment: Make the nose and mouth change when you click them.

Default Arguments

Let's suppose you have some movie clips on the stage and you would like to change the size at runtime. An easy way to do this is to write a function that will receive the movie clip and the new size.

```
resizeMC(star,1,.5);
resizeMC(ball,1.5,1);
function resizeMC(mc:MovieClip, wPercent:Number, hPercent:Number): void {
    mc.width*=wPercent;
    mc.height*=hPercent;
} //resizeMC
```

The star gets a new size that has the same width and is half as high. The ball will have a new size that is 1.5 times as wide and the same height. If your application only changes 1 dimension in most case, you could give a default value to the variables.

```
function resizeMC(mc:MovieClip, wPercent:Number=1, hPercent:Number=1):
void {
    mc.width*=wPercent;
    mc.height*=hPercent;
} //resizeMC
```

Now you can call the function the same way you did before or omit the 1 as an argument:

```
resizeMC(ball,1.5); //1.5 times as wide, same height
resizeMC(star); //no change
```

Note: You can not use a comma as a place holder like you can in some languages:

```
resizeMC(ball, ,2); will not work.
```

Return a value

Functions can return a value. In the example below the function returns the area of a rectangle:

```
var num:Number= area(4,3);
trace(num);
function area(w:int, h:int): int {
    return w*h;
} //area
```

When you test this it will display 12 in the output window. *If a function returns a value you can call the function without using the return value.*

Experiment: Write a function to make a movie clip rotate. Make a star and a square movie clip and rotate them when they are clicked. Change the movie so that all of the movie clips rotate each frame. Pass different values for each movie clip so that they rotate at different speeds.